Problem:
There are n processes in a queue. Each process has namei and timei. The round-robin scheduling handles the processes in order. A round-robin scheduler gives each process a quantum (a time slot) and interrupts the process if it is not completed by then. The process is resumed and moved to the end of the queue, then the scheduler handles the next process in the queue.

For example, we have the following queue with the quantum of 100ms.

A(150) - B(80) - C(200) - D(200)

First, process A is handled for 100ms, then the process is moved to the end of the queue with the remaining time (50ms).

B(80) - C(200) - D(200) - A(50)

Next, process B is handled for 80ms. The process is completed with the time stamp of 180ms and removed from the queue.

C(200) - D(200) - A(50)

Your task is to write a program which simulates the round-robin scheduling.

Input

n q

name1 time1

name2 time2

...

namen timen

In the first line the number of processes n and the quantum q are given separated by a single space.

In the following n lines, names and times for the n processes are given. namei and timei are separated by a single space.

Output

For each process, prints its name and the time the process finished in order.

Constraints

1 ≤ n ≤ 100000

1 ≤ q ≤ 1000

1 ≤ timei ≤ 50000

1 ≤ length of namei ≤ 10

1 ≤ Sum of timei ≤ 1000000

Sample Input 1

5 100
p1 150
p2 80
p3 200
p4 350
p5 20

Sample Output 1

p2 180
p5 400
p1 450
p3 550
p4 800

Notes

Template in C

Solution in Haskell (judge verdict: Accepted)

import Control.Applicative
import Control.Monad
import Control.Monad.State

main = do
        [n, q] <- map read.words<$>getLine
        pstrs  <- replicateM n getLine
        let procs = setProcess pstrs
        simulateProcess q procs

simulateProcess :: Int -> Queue Process -> IO ()
simulateProcess q = evalStateT (evalStateT (roundRobin q) 0)

type Queue a = ([a],[a])

enQueue :: Monad m => a -> StateT (Queue a) m ()
enQueue x = get >>= \(as,xs) -> put ((as, x:xs))

deQueue :: Monad m => StateT (Queue a) m a
deQueue = do
        (as, xs) <- get
        if null as then do
                       let xs' = reverse xs
                       put (tail xs', [])
                       return $ head xs'
                   else do
                       put (tail as, xs)
                       return $ head as

isEmpty :: Monad m => StateT (Queue a) m Bool
isEmpty = do
        (as,xs) <- get
        return $ null as && null xs


data Process = P { name :: String
                 , time :: Int } deriving (Show)

process :: [String] -> Process
process [x,y] = P x (read y)

setProcess :: [String] -> Queue Process
setProcess pstrings = execState action ([],[])
        where ps = map (process.words) $ pstrings
              action = sequence_ $ map enQueue ps

roundRobin :: Int -> StateT Int (StateT (Queue Process) IO) ()
roundRobin q = do
        t <- get
        p <- lift deQueue
        runProcess q p t
        emptyCheck
        where emptyCheck = do
                empty <- lift isEmpty
                if empty then return ()
                         else roundRobin q

runProcess :: Int -> Process -> Int -> StateT Int (StateT (Queue Process) IO) ()
runProcess q p t
    | q >= time p = do
        liftIO (putStrLn $ unwords [name p, show (t+time p)])
        modify (+ time p)
    | otherwise   = do
        let p' = p {time = time p - q}
        modify (+ q)
        lift $ enQueue p'